
Задача A. Строки в книге

Подсчитаем сначала, сколько страниц будет заполнено целиком, если мы напечатаем только первые N строк. Для этого разделим N на K с остатком. Неполное частное как раз и будет равно искомому числу – количеству страниц, на которых напечатано по K строк. Если остаток от деления равен нулю, то строка с номером N окажется последней на странице с номером N div K. Если же остаток отличен от нуля, то эта строка окажется уже на следующей странице – на странице с номером (N div K) + 1 – и ее порядковый номер будет в точности равен этому остатку – N mod K. Осталось написать короткую программу:

read(K,N);

if N mod K = 0 then write(N div K, ’ ’, K) 

else write (N div K + 1, ’ ’, N mod K);

Можно решить эту задачу и без использования условного оператора:

read(K,N);
write((n-1) div k) + 1, (n-1) mod k + 1);
Проверьте самостоятельно, что эти формулы дают тот же результат.

Задача B. Палиндромная последовательность

Начнем с функции, которая будет проверять, является ли последовательность a[1], a[2], …, a[k] палиндромной. Для этого нужно сравнить числа a[1] и a[k], a[2] и a[k-1], и т. д. до a[k div 2] и a[k+1-(k div 2)]. 
	a[1]
	a[2]
	…
	a[k div 2]
	a[k+1​​​-(k div 2)]
	…
	a[k-1]
	a[k]


Если в какой-то из пар числа окажутся неравными, то последовательность палиндромной не является. Напишем соответствующий код:

function palindrom (k : integer) : boolean;

var

  i : integer;

begin

palindrom:=TRUE;

  for i:=1 to k div 2 do
    if a[i]<>a[k+1-i] then begin

      palindrom := false;

      break;

    end;

end;

Теперь вернемся к исходной задаче. Заметим, что в худшем случае нам придется дописать N-1 чисел. Например, если дана последовательность 1 2 3 … N-1 N, то кратчайшая палиндромная последовательность будет выглядеть так: 1 2 3 … N-1 N N-1 … 3 2 1. В самом лучшем случае наша последовательность уже является палиндромной, и дописывать ничего не нужно. Таким образом, ответом в задаче является число от 0 до N-1.

Опишем один из алгоритмов, решающих нашу задачу. Для начала проверим, является ли данная нам последовательность палиндромной, используя вызов функции palindrom(N). Если нет, то проверим, достаточно ли добавить одно число. Ясно, что для того, чтобы получилась палиндромная последовательность, необходимо в конец дописать число a[1] : 

  a[N+1]:=a[1];

После этого опять проверим, получился ли палиндром, вызывая функцию palindrom(N+1). Если нас вновь не постиг успех, допишем в конец исходной последовательности два числа:

  a[N+2]:=a[1]; a[N+1]:=a[2] 

И т. д. до тех пор, пока функция palindrom на очередном шаге не вернет результат TRUE.

Покажем, как организовать этот процесс, используя цикл for:

 [1] for i:=0 to N-1 do begin

 [2]   for j:=1 to i do

 [3]     a[N+j]:=a[i+1-j];  {Добавляем в конец i чисел}

 [4]   if palindrom(N+i) then begin

 [5]     writeln(i); {Печатаем количество добавленных чисел}

 [6]     for j:=1 to i do

 [7]       write(a[N+j],’ ’);  {Печатаем добавленные числа}

 [8]   exit;

 [9]   end;

[10] end;

Обратите внимание, что мы принудительно завершаем работу программы оператором exit, если при очередной итерации цикла получается палиндромная последовательность. Можно было обойтись без этого, используя цикл while.

Теперь попробуем несколько модифицировать приведенный алгоритм. Дело в том, что мы выполняем лишние операции, дописывая в конец те же числа, что стоят в начале последовательности и затем сравнивая эти числа в функции palindrom. Ясно, что достаточно проверять лишь числа, которые входят в исходную последовательность. 

Для этого в цикле внутри функции palindrom нужно начинать проверять числа не с a[1], а с числа, которое сравнивается с числом a[N], то есть с числа a[k+1-N]:   

  for i:=k+1-N to k div 2 do ...

Тогда и в основной программе нет нужды дополнять массив новыми числами: строки [2] и [3] можно удалить, а строку [7] заменить такой: write(a[i+1-j],’ ’);
Отметим, что максимальная возможная длина палиндрома - 199. Это нужно учитывать при объявлении массива a.

Задача С. Кинотеатр

Заметим, что задача не всегда имеет решение: например, если мальчиков слишком много, а девочек мало (или, и того хуже, нет вовсе!), то посадить каждого мальчика рядом с девочкой не удастся. Поскольку рядом с каждой девочкой могут сидеть не более двух мальчиков, то количество мальчиков не может превосходить количества девочек более чем вдвое. То есть, если X>2Y, то задача решения не имеет. Также задача не имеет решения и в случае Y>2X.

Покажем, что в остальных случаях школьников можно рассадить требуемым образом. Если мальчиков и девочек поровну, то одна из искомых рассадок очевидна: BGBG…BG.

Пусть мальчиков больше, чем девочек, на k человек: X=Y+k. Тогда можно рассадить школьников так:

В тройках – 2k мальчиков и k девочек, в парах – Y-k мальчиков и Y-k девочек. Итого: 2k+(Y-k)=Y+k=X мальчиков, k+(Y-k)=Y девочек. Поскольку X<=2Y, то k<=Y, то есть Y-k>=0. Таким образом, предложенная рассадка всегда возможна. Случай, когда девочек больше, чем мальчиков, разбирается аналогично.

Теперь напишем программу.

read(X,Y);

if (X>2*Y) or (Y>2*X) then

  write(’NO SOLUTION’)

else if X>=Y then begin

       k:=X-Y;

       for i:=1 to k do

         write(’BGB’);

       for i:=1 to Y-k do

         write(’BG’);

     end

     else begin

       k:=Y-X;

       for i:=1 to k do

         write(’GBG’);

       for i:=1 to X-k do

         write(’GB’);

          end;

Обратим внимание на некоторые особенности работы этой программы. Если Y=X, то k=0 и первый цикл for не выполняется ни разу: все школьники рассаживаются парами “мальчик-девочка”. Если же X=2Y (или Y=2X), не выполняется второй цикл, поскольку Y-k (X-k, соответственно) равно нулю. В этом случаи все школьники рассаживаются тройками.

Задача D. Количество слов

В каждом слове есть первая буква. Поэтому вместо того, чтобы считать слова, можно считать первые буквы слов. Какая буква является первой буквой слова? Любая, слева от которой стоит не буква. Исключение составляет буква, слева от которой стоит символ “​-“, а перед ним - буква. Осталось написать программу, которая перебирает все символы строки, и считает, сколько среди них первых букв.

read(s);

for i:=1 to length(s) do

  if (s[i] in [‘a’..’z’,’A’..’Z’])  {s[i] – буква     }

   and not(s[i-1] in [’a’..’z’,’A’..’Z’]) 
{s[i-1] – не буква}

   and not ((s[i-1]=’-‘) and (s[i-2] in 
[‘a’..’z’,’A’..’Z’]))  {s[i-1] – не дефис}

   then n := n + 1; 

write(n);

У этой программы есть небольшой дефект. Когда мы проверяем, например, символ s[1], нам приходится обращаться к несуществующим символам s[0] и s[–1]. Чтобы избежать этого, можно изначально добавить в начало строки два символа, которые не повлияют на количество слов в строке, например, s:=’!!’+s, и начинать искать первые буквы слов, начиная с третьего символа строки:  for i:=3 to length(s) do ...

Задача E. Метро

Идею решения этой задачи можно кратко описать так. Рассмотрим все линии, на которых расположена станция A. До любой станции на любой из этих линий можно добраться со станции A без пересадок. Пометим все эти линии числом 0. Теперь рассмотрим все непомеченные линии, у которых есть общие станции с линиям, помеченными числом 0. До любой станции на любой из этих линий можно добраться со станции A, используя одну пересадку, но нельзя добраться без пересадок. Пометим эти линии числом 1. Далее, аналогично, рассмотрим все непомеченные линии, у которых есть общие станции с линиями, помеченными числом 1. На любую станцию этих линий можно попасть, используя две пересадки, поэтому пометим их числом 2, и т. д. до тех пор, пока мы впервые не пометим одну из линий, на которых есть станция B. Число, которым помечена эта линия, и будет являться ответом в нашей задаче. (Такой алгоритм называют волновым).
Теперь займемся реализацией этого алгоритма. Проще всего написать программу, используя структуру данных множество (set). Пусть l[i] - множество станций метро, лежащих на i-й линии (то есть l - массив множеств), current - множество станций, на которых можно оказаться, используя не более step пересадок. Построим множество next станций, на которых можно оказаться, используя не более (step+1) пересадок. Во-первых, все станции из множества current входят во множество next. Во-вторых, если на какую-то станцию можно попасть за не более чем step пересадок, и эта станция принадлежит в том числе и линии i, то тогда на любую станцию линии i можно попасть за не более чем (step+1) пересадок. То есть, если множество current* l[i] не пусто (звездочкой обозначено пересечение множеств, то есть их общая часть), все элементы множества l[i] нужно включить во множество next:

  next:=next+l[i]; {объединение множеств}

Осталось заметить, что если метро состоит из M линий, то либо со станции A на станцию B можно попасть не более чем с (M-1)-й пересадкой, либо нельзя попасть вовсе (A и B лежат в разных компонентах связности).

Теперь напишем программу.

{Чтение данных из входного файла}
  read(N,M);

  for i:=1 to M do begin

    read(k);

    for j:=1 to k do begin

      read(t);

      Include(l[i],t);   {Добавляем станцию t }

 {в множество l[i]    }

    end;

  end;

  read(A,B);

{Инициализация}

  emptyset:=[];  {Пустое множество}

  step:=0;

  current:=[A];



{Основной цикл}

  for step:=0 to M-1 do begin

    next:=current;

    for i:=1 to M do

      if current*l[i]<>emptyset then
        next:=next+l[i];

    if B in next then begin

      write(step); exit;

    end;

    current:=next;

  end;

{Пути от станции A до станции B нет}
  write(-1);

